

**Z1.** Dati RDF graf predstavljen u JSON-LD sintaksi potrebno je grafički predstaviti u skladu sa RDF notacijom. **(5 poena)**

```
{
  "@context": "http://schema.org",
  "@type": "JobPosting",
  "@id": "http://example.com/person/ab12",
  "baseSalary": "100000",
  "datePosted": "2011-10-31",
  "industry": "Computer Software",
  "jobLocation": {
    "@type": "Place",
    "address": {
      "@type": "PostalAddress",
      "@id": "http://example.com/person/cd34",
      "addressLocality": "Kirkland",
      "addressRegion": "WA"
    }
  },
  "salaryCurrency": "USD",
  "title": "Software Engineer"
}
```

**Z2.** Kreirati sledeće SPARQL upite nad RDF modelom datim u prethodnom zadatku:

- Upit kojim se ispisuje naziv (schema:title) radnog mesta oglasa (schema:JobPosting) kompanije čija je adresa (schema:address) u gradu 'Kirkland' (schema:addressLocality). **(2 poena)**
- Upit kojim se proverava da li postoji oglas za radno mesto (schema:JobPosting) u industriji (schema:industry) 'Computer Software' za koji je predviđena plata (schema:baseSalary) veća od 80000 USD (schema:salaryCurrency). **(3 poena)**

## Rešenja SPARQL upita<sup>1</sup>

```
PREFIX schema: <http://schema.org/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
SELECT ?title
WHERE {
    ?jobPost rdf:type schema:JobPosting .
    ?jobPost schema:title ?title .
    ?jobPost schema:jobLocation ?location .
    ?location rdf:type schema:Place .
    ?location schema:address ?address .
    ?address rdf:type schema:PostalAddress .
    ?address schema:addressLocality "Kirkland" .
}
```

```
PREFIX schema: <http://schema.org/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
ASK {
    ?jobPost rdf:type schema:JobPosting .
    ?jobPost schema:industry "Computer Software".
    ?jobPost schema:salaryCurrency "USD" .
    ?jobPost schema:baseSalary ?salary
        FILTER(xsd:int(?salary) > "80000"^^xsd:int) .
}
```

---

<sup>1</sup> Definicija prefiksa (tj. PREFIX elementi upita) je navedena radi kompletности rešenja, na ispitu nije neophodno definisati prefikse

**Z1.** Dati RDF graf predstavljen u JSON-LD sintaksi potrebno je grafički predstaviti u skladu sa RDF notacijom. **(5 poena)**

```
{
  "@context": {
    "schema": "http://schema.org/"
  },
  "@graph": [
    {
      "@id": "http://example.com/productsOffer9882454",
      "@type": "schema:Offer",
      "schema:itemCondition": "NewCondition",
      "schema:price": "281.98",
      "schema:priceCurrency": "USD"
    },
    {
      "@id": "http://example.com/productsWashburn9882454",
      "@type": "schema:Product",
      "schema:aggregateRating": {
        "@type": "schema:AggregateRating",
        "schema:ratingValue": "4.4",
        "schema:reviewCount": "15"
      },
      "schema:brand": "Washburn",
      "schema:model": "WD20S",
      "schema:name": "Washburn - WD20 Series 6-String Acoustic Guitar - Natural",
      "schema:offers": {
        "@id": "http://example.com/productsOffer9882454"
      }
    }
  ]
}
```

**Z2.** Kreirati sledeće SPARQL upite nad RDF modelom datim u prethodnom zadatku (nije potrebno pisati namespace-ove):

- Upit kojim se ispisuju cene (schema:price) svih ponuda (schema:Offer) proizvoda (schema:Product) koji pripadaju brendu (schema:brand) "Washburn". **(3 poena)**
- Upit kojim se proverava da li postoji ponuda (schema:offers) za proizvod (schema:Product) čija prosečna ocena (schema:AggregateRating) ima vrednost (schema:ratingValue) veću od 2. **(2 poena)**

## Rešenja SPARQL upita<sup>2</sup>

```
PREFIX schema: <http://schema.org/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
SELECT ?price
WHERE {
    ?product rdf:type schema:Product .
    ?product schema:brand "Washburn" .
    ?product schema:offers ?offer .
    ?offer rdf:type schema:Offer .
    ?offer schema:price ?price .
}
```

```
PREFIX schema: <http://schema.org/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
ASK {
    ?offer rdf:type schema:Offer .
    ?product rdf:type schema:Product .
    ?product schema:offers ?offer .
    ?product schema:aggregateRating ?rating .
    ?rating rdf:type schema:AggregateRating .
    ?rating schema:ratingValue ?rValue
        FILTER (xsd:decimal(?rValue) > "2"^^xsd:decimal)
}
```

---

<sup>2</sup> Definicija prefiksa (tj. PREFIX elementi upita) je navedena radi kompletenosti rešenja, na ispitu nije neophodno definisati prefikse

**Z1.** Dati RDF graf predstavljen u JSON-LD sintaksi potrebno je grafički predstaviti u skladu sa RDF notacijom. **(5 poena)**

```
{
  "@context": "http://schema.org",
  "@type": "Event",
  "location": {
    "@type": "Place",
    "address": {
      "@type": "PostalAddress",
      "addressLocality": "Denver",
      "postalCode": "80209",
      "streetAddress": "7 S. Broadway"
    },
    "name": "The Hi-Dive"
  },
  "name": "Typhoon with Radiation City",
  "offers": {
    "@type": "Offer",
    "price": "13.00",
    "priceCurrency": "USD",
  },
  "startDate": "2013-09-14T21:30"
}
```

**Z2.** Kreirati sledeće SPARQL upite nad RDF modelom datim u prethodnom zadatku (nije potrebno pisati namespace-ove):

- Upit kojim se ispisuju nazivi (*schema:name*) događaja (*schema:Event*) koji se održavaju u Denveru (*schema:addressLocality*) i čija je cena (*schema:price*) manja od 15 američkih dolara (USD). **(3 poena)**
- Upit kojim se proverava da li na lokaciji (*shema:Location*) pod imenom (*schema:name*) "The Hi-Dive" ima nekih dešavanja (*schema:Event*) u periodu koja počinju (*schema:startDate*) između 10. decembra 2016 (2016-12-10) i 24. decembra 2016. (2016-12-24). **(2 poena)**

### Rešenja SPARQL upita<sup>3</sup>

```
PREFIX schema: <http://schema.org/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
SELECT ?eventName
WHERE {
    ?event rdf:type schema:Event .
    ?event schema:name ?eventName .
    ?event schema:location ?location .
    ?location rdf:type schema:Place .
    ?location schema:address ?address .
    ?address rdf:type schema:PostalAddress .
    ?address schema:addressLocality "Denver" .
    ?event schema:offers ?offer .
    ?offer rdf:type schema:Offer .
    ?offer schema:priceCurrency "USD" .
    ?offer schema:price ?price FILTER( xsd:decimal(?price) < "15"^^xsd:decimal ).
}
```

```
PREFIX schema: <http://schema.org/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
ASK {
    ?event rdf:type schema:Event .
    ?event schema:location ?location .
    ?location rdf:type schema:Place .
    ?location schema:name "The Hi-Dive" .
    ?event schema:startDate ?eventDate
    FILTER (xsd:date(?eventDate) >= "2016-12-10"^^xsd:date
            && xsd:date(?eventDate) <= "2016-12-24"^^xsd:date)
}
```

---

<sup>3</sup> Definicija prefiksa (tj. PREFIX elementi upita) je navedena radi kompletности rešenja, na ispitu nije neophodno definisati prefikse